

**Supplementary data 5.** Community model JAGS code used in the analysis.

```
model {  
  
  # Priors  
  omega ~ dunif(0,1)  
  
  # Prior for species specific effects on occupancy and detection  
  for(k in 1:(M)){  
    lpsi[k] ~ dnorm(mu.lpsi, tau.lpsi) # Hyperparams describe community  
    betalpsi1 [k] ~ dnorm(mu.betalpsi1, tau.betalpsi1) # cat  
    betalpsi2 [k] ~ dnorm(mu.betalpsi2, tau.betalpsi2) # dog  
    betalpsi3 [k] ~ dnorm(mu.betalpsi3, tau.betalpsi3) # livestock  
  
    lp[k] ~ dnorm(mu.lp, tau.lp)  
    betalp1 [k] ~ dnorm(mu.betalp1, tau.betalp1) #only for date as detection covariate  
  }  
  
  # Hyperpriors  
  # For the model of occupancy  
  mu.lpsi ~ dnorm(0,0.01)  
  tau.lpsi <- pow(sd.lpsi, -2)  
  sd.lpsi ~ dunif(0,5)  
  
  mu.betalpsi1 ~ dnorm(0, 0.01)  
  tau.betalpsi1 <- pow(sd.betalpsi1, -2)  
  sd.betalpsi1 ~ dunif(0,3)
```

```

mu.betalpsi2 ~ dnorm(0, 0.01)
tau.betalpsi2 <- pow(sd.betalpsi2, -2)
sd.betalpsi2 ~ dunif(0,3)

mu.betalpsi3 ~ dnorm(0, 0.01)
tau.betalpsi3 <- pow(sd.betalpsi3, -2)
sd.betalpsi3 ~ dunif(0,3)

# for the model of detection
mu.lp ~ dnorm(0,0.01)
tau.lp <- pow(sd.lp, -2)
sd.lp ~ dunif(0,3)

mu.betalp1 ~ dnorm(0,0.01)
tau.betalp1 <- pow(sd.betalp1, -2)
sd.betalp1 ~ dunif(0,3)

# Superpopulation process: Ntotal species sampled out of M available
for(k in 1:(M)){
w[k] ~ dbern(omega)
}

# Likelihood
# Ecological model for true occurrence (process model), occupancy
for (k in 1:M){
for (i in 1:nsite) {
logit(psi[i,k]) <- lpsi[k] + betalpsi1[k] * cat[i] + betalpsi2[k] * dog[i] + betalpsi3[k] * livestock [i]
mu.psi[i,k] <- w[k] * psi[i,k]
z[i,k] ~ dbern(mu.psi[i,k])
}
}
}

```

```

# Observation model for replicated detection/nondetection observations
for(k in 1:(M)){
  for (i in 1:nsite){
    for(j in 1:nrep){
      logit(p[i,j,k]) <- lp[k] + betalp1[k] * date[i,j]
      p.eff[i,j,k] <- z[i,k] * p[i,j,k]
      Y[i,j,k] ~ dbern(p.eff[i,j,k])
      ### generate new data from model under consideration
      new.y[i,j,k] ~ dbern(p.eff[i,j,k])

      ### calculate Freeman-Tukey residuals for real and new data
      res[i,j,k] <- (Y[i,j,k] - sqrt(p.eff[i,j,k]))^2
      new.res[i,j,k] <- (new.y[i,j,k] - sqrt(p.eff[i,j,k]))^2
    }
    ###sum residuals over occasions
    R1[i,k] <- sum(res[i, ,k])
    new.R1[i,k] <- sum(new.res[i, , k])
  }
  ###sum residuals over sites
  R2[k] <- sum(R1[, k])
  new.R2[k] <- sum(new.R1[, k])
}

###sum residuals over (observed) species (all species were observed)
R3 <- sum(R2[])
new.R3 <- sum(new.R2[])

# Derived quantities
for(k in 1:(M)){
  occ.fs[k] <- sum(z[,k])      # Number of occupied sites among the 171

```

```
}  
for (i in 1:nsite) {  
  Nsite[i] <- sum(z[i,])      # Number of occurring species at each camera site  
}  
n0 <- sum(w[(nspec+1):(nspec+nz)]) # Number of unseen species  
Ntotal <- sum(w[])          # Total community size  
  
}
```